

PROJET CASTEM

**C**  
**A**  
**S**  
**T**  
**E**  
**M**

Modélisation d'in silo a gra  
**Modélisation d'in silo a grain**



## **INTRODUCTION**

Au cours de notre quatrième année de l'ISA-BTP, nous avons suivi une formation théorique d'environ 20 heures portant sur les éléments finis en mécanique. Afin de mettre à profit ce cours, nous avons appris à nous servir d'un code de calcul qui a été développé par le Département de Mécanique et de Technologie (DMT) du Commissariat français à l'Energie Atomique (CEA) : CASTEM 2000.

En effet, ce code de calcul permet de réaliser l'analyse de structures par la méthode des éléments finis, et permet également la modélisation en mécanique des fluides. Le développement de celui-ci entre dans le cadre d'une activité de recherche dans le domaine de la mécanique; le but étant de définir un instrument de haut niveau, pouvant servir de support valable pour la conception, le dimensionnement et l'analyse de structures et de composants, dans le domaine nucléaire comme dans le secteur industriel classique.

Ainsi, CASTEM 2000 se présente comme un système complet, intégrant non seulement des fonctions de calcul proprement dites, mais également des fonctions de construction du modèle et de traitement des résultats. Enfin, le principal but du code est de résoudre les problèmes non linéaires : plasticité, flambage, fluage, analyses sismiques, thermoplasticité, mécanique de la rupture, post-flambage, ruine des structures.

Dans cette optique, grâce à ce code de calcul, nous avons procédé à l'analyse d'une structure 3D s'inspirant d'un cas concret : un silo à grains.


En conséquence, notre projet de recherche s'articulera autour de trois parties principales. Dans un premier temps, il exposera les procédures mises en œuvre pour la modélisation de ce silo et évoquera les différents problèmes rencontrés lors de celle-ci. Puis, dans un second temps, ce rapport présentera les hypothèses et les paramètres utilisés pour réaliser les calculs. Enfin, notre étude se terminera par une analyse et une interprétation des résultats obtenus sous ce code de calcul.

# 1<sup>ère</sup> Partie : MODELISATION DE LA STRUCTURE

## A. PRESENTATION DU SILO

### 1. Généralités

Implanté dans la ville de Port-La-Nouvelle (Cf. fig.1), à 25 km de la ville de Narbonne et 45 de Perpignan, ce silo aura pour fonction le stockage de céréales tels que les grains de maïs ou de blé.



*Figure 1: Plan de situation*

Du fait de sa situation géographique, une attention toute particulière sera portée sur les fortes pressions de vent dû à la présence de la Tramontane.

### 2. Caractéristiques géométriques

Ce silo est constitué de divers éléments dont les caractéristiques géométriques sont les suivantes :

- Un réservoir cylindrique d'une hauteur totale de 32 m, avec :
  - Rayon intérieur = 6,75 m
  - Rayon extérieur = 6.95 m
  - Epaisseur du réservoir = 0.20 m

Remarque importante : Notre réservoir présentera à sa base un musoir de forme conique qui sera inclinée d'un angle  $\alpha = 40^\circ$  par rapport à l'horizontale.

- Une dalle pleine circulaire recouvrant le dessus du réservoir, avec :
  - Rayon = 6,95 m
  - Epaisseur = 0,20 m

- 12 poteaux carrés disposés tous les 30° supportant le réservoir et la dalle. Chacun de ces poteaux possède les dimensions suivantes :
  - Longueur = 0.50 m
  - Largeur = 0.50 m
  - Hauteur = 7 m

Afin de vous faire une idée plus concrète de notre projet, voici une photo du site (Cf. fig.2) :



***Figure 2: Photo du site***

***Remarque importante :*** Du fait de la complexité de la prise en compte des efforts de vent sur l'ensemble d'une barre (8 silos), notre étude se limitera à un seul de ces silos.

## **B. PROCEDURE DE MODELISATION**

- Pour cette étude, la modélisation de cette structure en 3 Dimensions (Cf. fig.3) nous a amené à choisir les paramètres dimensionnels suivants :
  - \* OPTI DIME 3 ELEM cub8;
  - \*

- Définition de divers paramètres qui serviront à la réalisation des éléments du silo :

```

*
epfut=0.2;           Epaisseur du réservoir
eppoto=0.5;         Epaisseur des poteaux
htpoto=7;           Hauteur des poteaux
rfut=6.85;          Rayon moyen du réservoir
rversoir=0.36;     Rayon du versoir
r1fut=rfut*2;
r2fut=r1fut*3;
r3fut=r2fut*(-1);
esite=1.25;         Coefficient de site (ici le site est exposé, on est au voisinage de la
                    mer méditerranéenne)

```

- Création de 2 points définissant l'axe du silo :

```

*
axe1=0. 0. 0.;
axe2=0. 0. 5.;
*

```

- Elaboration d'un douzième de cuve (car 12 poteaux) :

```

*
httot=Htfut+4.5;
dens 0.3;
S1=rversoir 0. 0.;
dens 1;
S2=rfut 0. 4.5;
S3= rfut 0. Httot;
*

```

- Création d'un poteau :

```

*
hpot2=(4.5*eppoto)/rfut;
eppoto2=eppoto/2;
eppoto3=eppoto2*(-1);
eppoto4=eppoto*(-1);
hpot3=hpot2*(-1);
hpot4=(hpot2 + htpoto)*(-1);
S2=rfut 0. 4.5;
s4=S2 PLUS (eppoto4. 0. hpot3);
s5=S2 PLUS (0. 0. hpot4);
s6=S5 PLUS (eppoto4. 0. 0.);
lign25=DROI S2 S5;
lign56=DROI S5 S6;
lign64=DROI S6 S4;
lign42=DROI S4 S2;
lign=lign25 et lign56 et lign64 et lign42;
spot= surf (lign) 'PLANE';
elim spot 10e-6;
spot = spot COUL 'BLEU' ;
*

```

- Multiplication des poteaux par 12 :

```

*
repete boum1 11;
spotm1= spot TOUR 30 axe1 axe2;
spot=spotm1 et spot;
fin boum1;
elim spot lign56 10.e-6;
spot = spot COUL 'JAUNE' ;
*

```

- Réalisation de 12 lignes au niveau de la base des poteaux afin de pouvoir bloquer ces derniers :


```
*  
repete boupot 11;  
spot2= lign56 TOUR 30 axe1 axe2;  
lign56=spot2 et lign56;  
fin boupot;  
elim lign56 10.e-6;  
lign56 = lign56 COUL 'TURQ';  
*
```

- Création du réservoir :

```
*  
lign1=DROI S1 S2;  
lign2=DROI S2 S3;  
cuve=lign1 et lign2 ;  
surf1= cuve ROTA 30 axe1 axe2 ;  
surf3=surf1;  
repete boucuve 11;  
surf3 =surf3 ROTA 30 axe1 axe2;  
fin boucuve;  
elim surf3 1.e-6;  
surf3 = surf3 COUL 'VERT';  
*
```

- Création de la dalle en partie supérieure du silo :

```
*  
S7=0. 0. htot;  
S8=TOUR S3 -120 axe1 S7;  
S9=TOUR S8 -120 axe1 S7;  
chapo1=C 12 S3 S7 S9 C 12 S7 S8 C 12 S7 S3;  
elim chapo1 10e-2;  
chapo2 = SURF chapo1 'PLAN';  
elim chapo2 10.e-6;  
chapo2 = chapo2 COUL 'ROUGE' ;  
surftot= surf3 et chapo2 et spot;  
elim surftot 1e-6;  
elim surftot lign56 1.e-6;  
trace surftot;  
*
```



***Figure 3: Modélisation du silo***

## **C. LES DIFFICULTES RENCONTREES**

La modélisation de cette structure nous a amené à rencontrer un certain nombre de problèmes auxquels nous avons dû faire face. En voici les principaux :

➤ **Maillage de la dalle :**

Dans un premier temps, on avait envisagé de créer cette dalle en réalisant la rotation d'un segment autour de l'axe du silo. Mais, nous avons rencontré un problème au niveau du centre de la dalle car on s'est retrouvé avec une concentration de points trop importante. Ainsi, le fait d'éliminer les nœuds proches entraînait la formation d'un trou circulaire au centre de celle-ci.

En conséquence, nous avons réalisé un cercle de même diamètre que le réservoir avec pour centre, un point situé sur l'axe du silo et à hauteur du haut du réservoir.

➤ **Liaison entre le maillage du réservoir et la dalle :**

Il a fallu jouer sur la densité du maillage de la dalle supérieure afin d'assurer la liaison avec celui du réservoir.

- Liaison entre le maillage du réservoir et les poteaux :  
Pour la construction du réservoir, nous avons fait en sorte de créer un nombre suffisant de points afin de pouvoir se servir de ces points pour la construction des poteaux. On est sûr comme cela sur, que les maillages sont parfaitement assemblés.
- Création de la partie conique du réservoir :  
Cette partie oblique compose la liaison entre deux cercles de diamètre différents. Nous avons donc du faire évoluer la densité en fonction de la proximité du petit cercle car pour définir le petit cercle il nous faut de petits éléments.
- L'impossibilité de réaliser des poteaux en 3D :  
En effet, le blocage en pied des poteaux 3D s'avérant trop complexe, nous avons finalement abandonné le fait de mettre des volumes pour ces derniers. La raison de ce choix se justifie du fait que cela aurait compliqué la définition de la structure. Il aurait fallu faire des surfaces courbe à la base des poteaux par la création de quatre points et on aurait ainsi bloqué ces surfaces. Néanmoins, à l'interface des poteaux et du réservoir, la forme arrondie de ce dernier aurait entraîné des difficultés dans la liaison des deux maillages.

## 2<sup>ème</sup> Partie : LES HYPOTHESES ET LES PARAMETRES DE CALCULS

### A. CARACTERISTIQUES MECANQUES

Pour étudier le comportement des éléments de ce silo, on a émis les hypothèses suivantes :

- Comportement élastique
- Masse volumique  $\rho = 2500 \text{ kg/m}^3$
- Module d'Young  $E$  du béton = 30 MPa
- Coefficient de Poisson  $\nu = 0.2$

\*

```
mopoto=MODE spot 'MECANIQUE' elastique coq4 coq3;  
mapoto=MATE mopoto youn 30.E9 Nu 0.2 rho 2500 epai 0.5;  
mocuve=MODE surf3 'MECANIQUE' elastique coq4 coq3;  
macuve=MATE mocuve youn 30.E9 Nu 0.2 rho 2500 epai 0.2;  
mochapo=MODE chapo2 'MECANIQUE' elastique coq4 coq3;  
machapo=MATE mochapo youn 30.E9 Nu 0.2 rho 2500 epai 0.2;
```

```
motot= mochapo et mocuve et mopoto;
```

```
matot= machapo et macuve et mapoto;
```

\*

Rmq1 : Dans notre étude, nous sommes dans l'impossibilité de définir des caractéristiques plus complètes pour les éléments ( $I_z, I_y, \dots$ ) car nous avons des inerties non constantes ou des coques circulaires.

Rmq2 : Les différents éléments (poteaux, cuve et dalle) ont été définis avec des éléments de types COQ3/COQ4 car suivant la densité du maillage, le logiciel construit un maillage uniquement avec des éléments de types COQ4 ou avec deux types COQ3/COQ4.

### B. APPLICATION DU POIDS PROPRE

La mise en place de cet effort (Cf. fig.4) nécessite la création d'un champ de points sur l'ensemble du maillage. Ainsi, nous avons créé un champ sur lequel on appliquera l'accélération de pesanteur  $9.81 \text{ m.s}^{-2}$ . De là, on multiplie ce champ par la masse de notre structure.

\*

```
mass1=MASS motot matot;  
ch1=MANU chpo (surftot) 1 UZ -9.81;  
poids=mass1*ch1;
```

\*

### C. PRESSION DUE AU VENT

Pour pouvoir appliquer la pression du vent (Cf. fig.4) sur notre ouvrage, nous nous sommes servi d'une fonction de distribution angulaire qui correspond à une décomposition en série de FOURIER à dix termes d'une fonction proposée par les règles NV65. Cette méthode est issue d'une thèse réalisée par un chercheur de l'INSA de Lyon (Mr WITASSE).

➤ Définition des coefficients de FOURIER :

```
*
tabk=table;
tabk.0=-0.3483;
tabk.1=0.2864;
tabk.2=0.7562;
tabk.3=0.4081;
tabk.4=-0.05410;
tabk.5=-0.09369;
tabk.6=-0.0005660;
tabk.7=0.01485;
tabk.8=0.007298;
tabk.9=0.01111;
tabk.10=0.01287;
*
```

➤ Définition de la pression de l'air sur le réservoir :

```
*
x y z=COORD (surf3);
theta1=atg (Y / x);
M1=MASQUE x 'INFERIEUR' 0.;           On rajoute 180° pour les x<0
theta=theta1 + (180*M1);
th1=PROG 0. pas 1. 360.;
press1=tabk.0 * (cos (0. * th1));
repete bou1 10;
press1=press1 + (tabk . &bou1 * (cos (&bou1 * th1)));
fin bou1;
evp=evol MANU th1 press1;
ch1=manu chpo (surf3) 1 SCAL 1.;
th2=theta - ch1;
chp1=tabk.0 * (cos (0. * th2));
repete bou1 10;
chp1=chp1 + (tabk . &bou1 * (cos (&bou1 * th2)));
fin bou1;
```

*Nous avons appliqué un coefficient au vent permettant de faire varier celui-ci suivant la hauteur du silo (cf : NV65)*

```
z33= COORD 3 surf3;
hterre=2.5*(-1);
h33=h+10.5;
masq33=z33 MASQUE inferieur h33;
masq34=z33 MASQUE superieur hterre;
ch35=(2.5*((z33+18)/(z33+60)))*masq33*masq34;
```

*Nous avons pris une pression dynamique de base extrême de 210 daN/m<sup>2</sup> ( région III)*


```
FVENT=PRESSION 'COQU' (motot)((2100*esite)*chp1) 'NORM';
max1=MAXI fvent abs;
vecvent=vecteur fvent fx fy fz (1/max1) turq;
*
```

## D. EFFORTS DUS AU POIDS DU MAIS


Pour l'application des forces dues au maïs sur notre réservoir (Cf. fig.4), nous avons assimilé le maïs à de l'eau. Ceci a pour désavantage de négliger les forces de frottements interne du

mais, mais donne une bonne approximation de l'effet des grains sur la structure. Le maïs a pour masse volumique  $800\text{kg/m}^3$ .

```
*
romais=8000;
z1=COORD 3 surf3;
pr=romais*9.81;
chh=MANU chpo surf3 1 scal h;
masq1=z1 MASQUE inferieur h;
masq2=z1 MASQUE superieur hmin;
ch1=((z1-chh)*pr)*masq1*masq2;
mais=PRESSION 'COQU' mocuve ch1 'NORM';
max2=MAXI mais abs;
vecmais=VECTEUR mais fx fy fz (1/max2) rouge;
trac (vecmais et vecvent) surf3tot cach;
*
```



**Figure 4:** Efforts appliqués sur le silo (vent et mais)



Rmq(vent) : Sur les deux schéma ci dessus, nous avons décomposé les deux actions externes au silo a savoir l'action du vent et du mais. On remarque que l'action du vent est bien parabolique en fonction du point d'application sur le silo. On s'aperçoit également qu'en tête de silo, les nœuds subissent une action du vent moindre. Ceci est parfaitement compréhensible car le dernier nœud ne reprend que la moitié de la surface d'un nœud courant.

Rmq(mais) : La force due au mais est bien triangulaire car elle est proportionnelle a la hauteur du mais.

## 3<sup>ème</sup> Partie : LES RESULTATS OBTENUS ET LEUR ANALYSE

### A. VISUALISATION DES DEFORMATIONS DANS LA STRUCTURE

Dans un premier temps, nous avons voulu visualiser les déformations subies par le silo sous l'effet du poids du maïs, du poids propre de la structure et sous l'effet du vent (Cf. fig.5) :


\*

```

rig1=RIGI motot matot;
BLO1=BLOQ (lign56) DEPL ROTA;
RITOT=RIG1 ET BLO1;
DEP1=RESOU RITOT (mais et POIDS et FVENT);
def1=DEFO dep1 surftot;
sig1=SIGMA dep1 motot matot;   Calcul des contraintes dans le repère général
trac (sig1 )motot matot def1;

```


\*



***Figure 5: Déformations subies par la structure***

A partir de là, on a décidé de visualiser les contraintes de VON MISES (Cf. fig.6) sur la structure non déformée afin de connaître les « endroits » où notre silo était le plus sollicité :

```
*
VM1 = VMIS motot sig1 matot;
trace motot VM1 surftot;
*
```



***Figure 6: Contraintes de Von Mises***

Puis, on a cherché à extraire la contrainte maximale de VON MISES, ceci en vue d'une analyse plus précise de la structure :

```
*
Smax = MAXI sig1;
smax = smax / 1.E6;
*
MESS ' RESULTAT ';
MESS ' ----- ';
SAUT 1 LIGN;
MESS ' CONTRAINTE max = :! smax ' MPA';
```

```

*
$ * *****
$ * ** EXTRACTION DE LA CONTRAINTE maximale de Von mises
$ * **a faire avec Rtens**
$ *
$ *
$ * Smax = maxi sig1;
$ * smax = smax / 1.E6;
$ *
$ *
$ * MESS ' RESULTAT ' ;
RESULTAT
$ * MESS ' ----- ' ;
$ * SAUT 1 LIGN;

$ * MESS ' CONTRAINTE max = : ' smax ' MPA ' ;
CONTRAINTE max = : 20.751 MPA
*


```

Enfin, nous avons cherché les contraintes du silo dans le repère cylindrique (Cf. fig.7) car ce repère correspond

```

*
champ3= RTENS sig1 motot 'CYLI' s1 axe2;
trace champ3 motot matot;
*

```



**Figure 7 :** Contraintes dans le silo

On a fait de même qu'avec la contrainte de Von Mises, nous avons extrait la contrainte maximale dans la nouvelle base :

```

*****
* *tracé des contraintes dans un repère cylindrique
* champ3= rtens sig1 motot 'CYLI' s1 axe2;
menage automatique gain 309216 actifs 1632048 mots 238 segments aprs RTEN
* trace champ3 motot matot;
*
* *EXTRACTION DE LA CONTRAINTE maximale avec Rtens
* $max2 = maxi champ3;
* $max2 = $max2 / 1.E6;
* *
* *
* *
$ * MESS ' RESULTAT ' ;
RESULTAT
$ * MESS ' ----- ' ;
$ * SAUT 1 LIGN;

$ * MESS ' CONTRAINTE max = : ' $max2 ' MPA' ;
CONTRAINTE max = : 20.720 MPA
$ *
$ *

```

*Commentaires :* Les contraintes maximales sont proches mais pas égales, pour l'exploitation des résultats de la cuve, il nous faut utiliser les contraintes dans le repère cylindrique.

## B. VALIDITE DES RESULTATS

Afin d'avoir une idée sur la validité de nos résultats, nous avons demandé à CASTEM de calculer l'erreur globale lié au calcul des contraintes (sig1) et de nous indiquer les zones du silo où les erreurs seraient à priori les plus importantes (Cf. fig.8).


```

*
x1 cham2 =ERRE motot sig1 matot ;
trace cham2 motot;
*

```

Remarque : Cette erreur n'est qu'un indice car on ne peut pas faire un pourcentage par rapport à la valeur réelle étant donné que l'on ne connaît pas la valeur de celle-ci (Calcul aux éléments finis → toujours une approximation de résultats).

Cependant, cela permet à l'utilisateur de connaître les zones susceptibles de contenir des résultats moins précis et donc de réduire le maillage au voisinage de ces zones avec l'objectif d'obtenir des résultats plus corrects.



**Figure 8:** Estimation de l'erreur lors du calcul des contraintes

Commentaires: Les points où l'estimation de l'erreur est maximale sont les points qui sont généralement à l'interface de deux maillages non coplanaires. Il serait intéressant de réduire ou d'augmenter le maillage et de comparer les contraintes en plusieurs points (points choisis dans les zones où l'erreur est la plus importante) avec les deux maillages différents. On a comme cela une idée sur la précision de nos mesures et sur la validité des résultats.


### C. DEFORMÉE PAR RAPPORT A LA STRUCTURE INITIALE

On a superposé la structure initiale avec la structure déformée ce qui permet de mieux visualiser la déformation (Cf. fig.9).

\*

```
AMP1 = 1.E-7;
REA1 = REAC DEP1 RIG1;
VEC1 = VECT (mais et POIDS et FVENT) AMP1 FX FY FZ ROUGE;
VEC2 = VECT REA1 AMP1 FX FY FZ BLEU;
DEF0 = DEFO 0. DEP1 surfTOT;
DEF1 = DEFO 30. DEP1 surfTOT ( VEC1 ET VEC2 ) VERT;
TITRE ' DEFORMATION, CHARGEMENT, REACTION';
TRACE surfTOT (DEF0 ET DEF1);
```

\*




***Figure 9: Représentation du silo déformé***

## **D. REACTIONS D'APPUI A LA BASE DES POTEAUX**

En outre, on a voulu savoir les valeurs des réactions d'appuis au niveau de la base des poteaux (Cf. fig.10) :

```
*
R1=REAC ritot dep1;
r2=MAXI r1;
vecreac=VECTEUR r1 fx fy fz (1/r2) rouge;
trace vecreac lign56 cach;
list R1;
*
```

On a 6 composantes par nœud et 24 nœuds :



	MZ	FX	FY	FZ	MX	MY
E+00	8	-1.79051E+06	-3.49653E+04	3.30692E+06	4.80708E+04	0.00000
		6.39375E+03				
E+04	9	-1.54463E+06	9.34093E+05	3.34841E+06	-4.65732E+04	2.68890
		-7.64934E+03				
E+04	10	2.12636E+06	-1.22688E+06	9.55162E+06	-4.74770E+04	2.74108
		1.67725E+04				
E+05	11	-3.02494E+06	5.71803E+06	5.74918E+06	-2.36562E+05	4.09738
		-7.57719E+04				
E+05	12	4.34756E+06	-7.63067E+06	3.42102E+07	-2.41954E+05	4.19077
		1.50677E+05				
E+06	13	5.40382E+05	9.70459E+06	-1.25609E+07	-2.94697E-10	2.66481
		-6.08305E+05				
E+06	14	2.24375E+05	-1.31219E+07	5.12854E+07	-2.94583E-10	2.66377
		6.75773E+05				
E+06	15	7.99770E+06	1.07829E+07	-3.19810E+07	3.77336E+06	6.53565
		-1.71431E+06				
E+06	16	-8.45917E+06	-1.57591E+07	7.01415E+07	3.77496E+06	6.53842
		1.92996E+06				
E+06	17	1.32551E+07	5.18266E+06	-3.18787E+07	8.88373E+06	5.12902
		-2.31473E+06				
E+06	18	-1.57487E+07	-9.87945E+06	7.32656E+07	8.89341E+06	5.13462
		2.64199E+06				
E+06	19	1.27736E+07	-1.03943E+06	-2.14695E+07	4.93266E+06	4.34044
		-1.11006E+06				
E-09	20	-1.65705E+07	-3.14790E+05	6.57211E+07	4.93931E+06	4.34629
		1.27391E+06				
E+06	21	1.16439E+07	-5.50442E+06	-2.18529E+07	-4.36982E+06	2.52292
		1.13759E+06				
E+06	22	-1.42597E+07	8.61315E+06	6.60546E+07	-4.37503E+06	2.52592
		-1.30095E+06				
E+06	23	8.97704E+06	-1.12651E+07	-3.26767E+07	-5.15923E+06	8.93605
		2.33084E+06				
E+06	24	-8.80649E+06	1.66418E+07	7.41915E+07	-5.16438E+06	8.94497
		-2.65493E+06				
E+06	25	1.51540E+06	-1.35455E+07	-3.27256E+07	-8.30338E-10	7.53791
		1.71671E+06				
E+06	26	5.68915E+05	1.81613E+07	7.12502E+07	-8.30503E-10	7.53941
		-1.92301E+06				
E+06	27	-4.44611E+06	-8.70906E+06	-1.31435E+07	1.30749E+06	2.26463
		6.03519E+05				
E+06	28	6.81380E+06	1.13136E+07	5.16105E+07	1.30570E+06	2.26154
		-6.56084E+05				
E+05	29	-5.41687E+06	-3.37725E+06	5.37752E+06	3.83159E+05	2.21217
		7.28602E+04				
E+05	30	7.48641E+06	4.36710E+06	3.37630E+07	3.91177E+05	2.25846
		-1.38634E+05				
E+00	31	2.43557E+06	2.12079E+02	9.47638E+06	4.91305E+04	0.00000
		-1.53549E+04				

**Figure 10: Réactions d'appui à la base des poteaux**

Commentaires : Les valeurs ci dessus caractérisent les réactions d'appuis des encastremements de poteaux. On retrouve bien trois moments et trois forces, suivant les trois directions principales. Grâce à ces valeurs, on pourra éventuellement dimensionner les armatures en pied de poteaux.

## E. ETUDE DES CONTRAINTES EN TETE DE POTEAUX (AUX POINTS S4 ET S2)

Comme la zone la plus sollicitée de notre structure se trouvait proche de la tête des poteaux, on a calculé et extrait les valeurs des contraintes en S2 et S4 :

```
* Pour le point S4
sig99=REDU sig1 mopoto;
poto99=CALP sig99 mapoto mopoto 'SUPE';
CHSIG=CHANGER chpo mopoto poto99 ;
aaa= EXTR CHSIG SMXX S4;
bbb= EXTR CHSIG SMYY S4;
ccc= EXTR CHSIG SMZZ S4;
ddd= EXTR CHSIG SMXY S4;
eee= EXTR CHSIG SMYZ S4;
fff= EXTR CHSIG SMXZ S4;
MESS ' SMXX :' aaa ' MPA';
MESS ' SMYY :' bbb ' MPA';
MESS ' SMZZ :' ccc ' MPA';
MESS ' SMXY :' ddd ' MPA';
MESS ' SMYZ :' eee ' MPA';
MESS ' SMXZ :' fff ' MPA';
*

$ **
$ * * EXTRACTION DE LA CONTRAINTE SIGYY EN S4
$ * MESS ' SMXX :' aaa ' MPA';
SMXX : 7.3767 MPA
$ * MESS ' SMYY :' bbb ' MPA';
SMYY : 1.1867 MPA
$ * MESS ' SMZZ :' ccc ' MPA';
SMZZ : 0.0000 MPA
$ * MESS ' SMXY :' ddd ' MPA';
SMXY : -1.4453 MPA
$ * MESS ' SMYZ :' eee ' MPA';
SMYZ : -0.10496 MPA
$ * MESS ' SMXZ :' fff ' MPA';
SMXZ : 2.13696E-03 MPA
$ *
$ * * EXTRACTION DE LA CONTRAINTE SIGYY EN S2
$ * MESS ' SMXX :' aaa1 ' MPA';
SMXX : -8.0238 MPA
$ * MESS ' SMYY :' bbb1 ' MPA';
SMYY : -2.0858 MPA
$ * MESS ' SMZZ :' ccc1 ' MPA';
SMZZ : 0.0000 MPA
$ * MESS ' SMXY :' ddd1 ' MPA';
SMXY : -1.4623 MPA
$ * MESS ' SMYZ :' eee1 ' MPA';
SMYZ : 0.14761 MPA
$ * MESS ' SMXZ :' fff1 ' MPA';
SMXZ : -7.80310E-02 MPA
$ *
```

On a cherché les contraintes maximales suivant l'épaisseur des éléments avec la commande CALP :

```
CHAM2 = CALP sig1 matot motot 'SUPE';
CHAM3=MAXI CHAM2;
CHAM3=CHAM3/(10e6);
CHAM4 = CALP sig1 matot motot 'INFE' ;
CHAM5=MAXI CHAM4;
CHAM5=CHAM5/(10e6);
CHAM6 = CALP sig1 matot motot ;
CHAM7=MAXI CHAM6;
CHAM7=CHAM7/(10e6);
```

```

*
MESS ' RESULTAT ';
SAUT 1 LIGN;
MESS ' CONTRAINTE CALCULEE max cham3 :' CHAM3 ' MPA';
MESS ' CONTRAINTE CALCULEE min cham5 :' CHAM5 ' MPA';
MESS ' CONTRAINTE CALCULEE moy cham7 :' CHAM7 ' MPA';
*
$ *
$ * CHAM2 = CALP sig1 matot motot 'SUPE' ;
$ * CHAM3=maxi CHAM2;
$ * CHAM3=CHAM3/(10e6);
$ * CHAM4 = CALP sig1 matot motot 'INFE' ;
$ * CHAM5=maxi CHAM4;
$ * CHAM5=CHAM5/(10e6);
$ * CHAM6 = CALP sig1 matot motot ;
$ * CHAM7=maxi CHAM6;
$ * CHAM7=CHAM7/(10e6);
$ *
$ *
$ * MESS ' RESULTAT ' ;
RESULTAT
$ * SAUT 1 LIGN;

$ * MESS ' CONTRAINTE CALCULEE max cham3 :' CHAM3 ' MPA';
CONTRAINTE CALCULEE max cham3 : 19.538 MPA
$ * MESS ' CONTRAINTE CALCULEE min cham5 :' CHAM5 ' MPA';
CONTRAINTE CALCULEE min cham5 : 15.967 MPA
$ * MESS ' CONTRAINTE CALCULEE moy cham7 :' CHAM7 ' MPA';
CONTRAINTE CALCULEE moy cham7 : 10.376 MPA
$ *

```

Commentaires : Ces valeurs sont logiques, on remarque que la fibre moyenne est la moins sollicitée . Ceci surement du au fait qu'elle est la plus proche de la fibre neutre.


## G. VIBRATIONS

Pour terminer cette étude, et afin de profiter un peu plus des nombreuses fonctions de CASTEM, on a cherché à obtenir le mode de ruine de notre silo sous vibrations. Pour cela, on lui demandé de calculer six modes propres. On a fait débiter les fréquences à 0.001 Hz car les basses fréquences sont les plus préjudiciables pour les bâtiments (Cf. fig.11).

```

*
n2=6;
sol1=VIBRATION SIMULTANEE 0.001 6 ritot mass1;
REPETER bou1 n2;
n1=DIME sol1;
dep1=TIRE sol1 'DEPL' rang &bou1;
def1=DEFO dep1 (surftot);
On a également intégré une animation de la structure sous CASTEM
def2=ANIME 5 surftot dep1 &bou1 ;
trac def1 cach;
trac def2 oscil cach;
fin bou1;
*

```



***Figure 11: Silo soumis à des vibrations***

Commentaires:

On constate que notre structure fait une rotation (les poteaux subissent de la torsion) entraînant ainsi sa ruine. Ceci provient de l'absence de contreventement de celle-ci. Cependant, rappelons que notre étude s'est limité à un seul silo alors que dans la réalité celui-ci fait parti d'un ensemble composé de 8 silos liaisonné par l'intermédiaire de poutres situés en haut de ces derniers.

## **CONCLUSION**

Le cours de mécanique des milieux continus se finalise par l'utilisation du logiciel de calcul CASTEM. Nous avons pu découvrir et se familiariser avec un des outil le plus performant en matière de calcul des structures complexes (plaques et coques).

Malgré un cours très théorique tout comme l'utilisation de ce logiciel, nous avons choisi de rendre notre projet le plus concret possible. Tout d'abord par le sujet lui-même qui correspond à un ouvrage existant, puis en le situant dans son environnement réel ; concernant les paramètres de calcul, nous avons essayé de nous rapprocher le plus possible de la réalité en faisant notamment varier l'effet du vent selon sa hauteur et sa direction, et en lui imposant une valeur conforme à la réglementation NV65.

Concernant l'exploitation des résultats, le logiciel propose un grand nombre de fonctions, dans notre projet nous avons appliqué celles qui nous paraissaient les plus adaptées (Déformées, déformations , réactions, vibrations). Malgré tout l'approche de ces résultats reste très théorique et il reste très difficile de faire un lien entre les résultats donnés par castem et un règlement de calcul de structure. Nous pensons qu'approfondir l'étude du logiciel permettrait justement de faire ce lien, ce qui rendrait à son utilisation une touche de réalisme.

Le projet n'est pas fini, et il serait intéressant de l'approfondir en vu de réaliser un dimensionnement béton armé. Au final l'utilisation de ce logiciel reste une expérience intéressante et enrichissante quelque soit notre future carrière.